Solving an optical cavity using scattering transfer matrices[1]

The S-matrix (scattering matrix, S-parameters) is a common way to specify the amplitude reflection and transmission coefficients of a system (optical component, radio-frequency electronic device, quantum mechanical scattering scenario, etc). The S-matrix gives the amplitudes of waves scattering *out* of the system in terms of the amplitudes scattering *in*.

Suppose the system looks like this, with amplitude a_1 incident from the left, a_2 incident from the right, and b_1 and b_2 emitted to the left and right, respectively:



Then the S-matrix is defined as the operator which takes **a** and gives you **b**:

$$\begin{bmatrix} b_1 \\ b_2 \end{bmatrix} = S \begin{bmatrix} a_1 \\ a_2 \end{bmatrix}$$

For instance, possible[2] S-matrices for a mirror and for propagation through free space are:

$$S_{\text{mirror}} = \begin{pmatrix} r & it \\ it & r \end{pmatrix} \qquad S_{\text{space}} = e^{i\omega\frac{L}{c}} \begin{pmatrix} 0 & 1 \\ 1 & 0 \end{pmatrix}$$

The S matrix is easy to measure and to calculate, but sometimes it is convenient to use a different matrix, which gives the amplitudes on the left side of a component in terms of the amplitudes on the right side. This matrix is called the scattering transfer matrix, or simply the T-parameters or T-matrix. The nice property of the T matrix is that the T matrix for a sequence of components is simply the product of the T matrices for the individual components[3]. For example, to find the T-matrix for a cavity consisting of two mirrors separated by free space, you would simply multiply together the T-matrices for the first mirror, the free space, and the second mirror.

The T-matrix relates the amplitudes like this:

$$\begin{bmatrix} b_1 \\ a_1 \end{bmatrix} = T \begin{bmatrix} b_2 \\ a_2 \end{bmatrix}$$

The transformations between T and S are a little strange[4], kind of like half an inverse:

$$\begin{aligned} T_{11} &= -\det(S)/S_{21} \\ T_{12} &= S_{11}/S_{21} \\ T_{21} &= -S_{22}/S_{21} \\ T_{22} &= 1/S_{21} \end{aligned} \qquad \begin{aligned} S_{11} &= T_{12}/T_{22} \\ S_{12} &= (\det T)/T_{22} \\ S_{21} &= 1/T_{22} \\ S_{22} &= -T_{21}/T_{22} \end{aligned}$$

Applynig the $S \to T$ transformation to the S matrices for a mirror and for free space, we find:

---

[1]Tobin Fricke, 2010-08-29

[2]There is some freedom in the choice of phase, but you must have $\det S = 1$ for energy conservation or $\det S = 1 - L$ if an element has loss.

[3]It would seem that the T matrix is not so useful for branching structures, such as a Michelson interferometer.

[4]Is there a nicer, more abstract way to write the transformations between S and T?

$$T_{\text{mirror}} = -\frac{i}{t}\begin{pmatrix} -(r^2+t^2) & r \\ -r & 1 \end{pmatrix} \qquad T_{\text{space}} = \begin{pmatrix} e^{i\omega\frac{L}{c}} & 0 \\ 0 & e^{-i\omega\frac{L}{c}} \end{pmatrix}$$

Now we can compute the T matrix for a Fabry Perot cavity just by multiplying these guys together (this is the whole point!): $T_{\text{cavity}} = T_{\text{mirror1}} \cdot T_{\text{free space}} \cdot T_{\text{mirror2}}$. Assuming $r^2 + t^2 = 1$ (lossless optics) due to laziness, and letting $\phi \equiv \omega L/c$ for brevity:

$$\begin{aligned} T_{\text{cavity}} &= \frac{-1}{t_1 t_2}\begin{pmatrix} -1 & r_1 \\ -r_1 & 1 \end{pmatrix}\begin{pmatrix} e^{i\omega\frac{L}{c}} & 0 \\ 0 & e^{-i\omega\frac{L}{c}} \end{pmatrix}\begin{pmatrix} -1 & r_2 \\ -r_2 & 1 \end{pmatrix} \\ &= \frac{-1}{t_1 t_2}\begin{pmatrix} e^{i\phi} - e^{-i\phi} r_1 r_2 & e^{-i\phi} r_1 - e^{i\phi} r_2 \\ e^{i\phi} r_1 - e^{-i\phi} r_2 & e^{-i\phi} - e^{i\phi} r_1 r_2 \end{pmatrix} \end{aligned}$$

Now, to get the cavity reflectivity and transmission coefficients, we transform the whole thing back to S and extract S_{11} and S_{21}:

$$r_c \equiv S_{11} = \frac{e^{-i\phi} r_1 - e^{i\phi} r_2}{e^{-i\phi} - e^{i\phi} r_1 r_2} = \frac{r_1 - r_2 e^{i2\phi}}{1 - r_1 r_2 e^{i2\phi}}, \quad t_c \equiv S_{21} = \frac{-t_1 t_2}{e^{-i\phi} - r_1 r_2 e^{i\phi}} = \frac{-t_1 t_2 e^{i\phi}}{1 - r_1 r_2 e^{2i\phi}}$$

which are, of course, the usual results.

The above calculations are also computed in the attached Mathematica program (on the following page).

resonant cavity using scattering transfer matrices

Here I define two functions which convert between S - parameters and T - parameters. These functions only work for 2×2 matrices; I don' t know the general form of the transformation nor whether it really makes sense to use T - parameters for devices with more than two ports (maybe 2N ports?).

Tobin Fricke 2010-08-29

In[1]:=
```
StoT[S_] := {{-Det[S] / S[[2, 1]], S[[1, 1]] / S[[2, 1]]}, {-S[[2, 2]] / S[[2, 1]], 1 / S[[2, 1]]}}
```

In[2]:=
```
TtoS[T_] := {{T[[1, 2]] / T[[2, 2]], Det[T] / T[[2, 2]]}, {1 / T[[2, 2]], -T[[2, 1]] / T[[2, 2]]}}
```

Make sure that the composition of these operations is the identity :

In[3]:=
```
With[{S = Array[s, {2, 2}]}, TtoS[StoT[S]] == S ]
```

Out[3]= True

Now I define the S - matrices for a lossless mirror and for free - space :

In[4]:=
```
Smirror = {{r, I t}, {I t, r}}
Sfreespace = Exp[I ϕ] {{0, 1}, {1, 0}}
```

Out[4]= $\{\{r, i\,t\}, \{i\,t, r\}\}$

Out[5]= $\{\{0, e^{i\phi}\}, \{e^{i\phi}, 0\}\}$

Convert the S - matrices to T - matrices :

In[6]:=
```
Tmirror = StoT[Smirror]
Tfreespace = StoT[Sfreespace]
```

Out[6]= $\left\{\left\{-\frac{i\left(-r^2-t^2\right)}{t}, -\frac{i\,r}{t}\right\}, \left\{\frac{i\,r}{t}, -\frac{i}{t}\right\}\right\}$

Out[7]= $\{\{e^{i\phi}, 0\}, \{0, e^{-i\phi}\}\}$

Define a Fabry - Perot cavity using the T - matrices :

In[8]:=
```
Tmirror1 = Tmirror /. {r → r1, t → t1}; Tmirror2 = Tmirror /. {r → r2, t → t2};
Tfp = Tmirror1 . Tfreespace . Tmirror2;
```

To extract the cavity reflection coefficient, we transform back to the S - matrix :

In[10]:=
```
Sfp = TtoS[Tfp];
```

In[11]:=
```
rc = FullSimplify[Sfp[[1, 1]]];
tc = FullSimplify[Sfp[[2, 1]]];
```

Check whether this is equal to the usual form of rc :

In[13]:=
```
rc == (r1 - (r1^2 + t1^2) r2 Exp[I 2 ϕ]) / (1 - r1 r2 Exp[I 2 ϕ]) // Simplify
```

Out[13]= True

It is.